Special Relativity

(Lorentz transformations)

Dr. Luigi E. Masciovecchio
email: luigi.masciovecchio@virgilio.it
(first published on http://mio.discoremoto.alice.it/luigimasciovecchio/, november 2009)
available as notebook and PDF on http://sites.google.com/site/luigimasciovecchio/
2017.06.19

```
Print["Document revision: ", IntegerPart[Date[]]]
```

```
Document revision: {2017, 6, 19, 7, 57, 44}
```

Dear Colleagues,

This is my personal *Mathematica* notebook on special relativity focussing on the **Lorentz transformations**. This document wasn't originally intended for publication, but a few formulas are maybe of interest to you, so here they are. The code seems to work well and I added some comments to make it more understandable. This is *not* an introduction to this field, so use it at your own risk! I have stolen many ideas (with some corrections) from Ladislau Radu "Herleitung der Lorentz-Transformation" (2006, Internet).

In Einstein's theory of **special relativity** (first published in 1905) the Lorentz transformation converts between two different inertial observers' measurements of space and time, where one observer is in (constant) motion with respect to the other. Here I present a derivation of the Lorentz transformation **without** the assumption that the speed of light is a constant. For more details about this approach see Jean-Marc Lévy-Leblond, "One more derivation of the Lorentz transformation", American Journal of Physics 44, 271-277 (1976).

1. Spezielle Lorentztransformation (Lorentz transformation for frames in standard configuration)

■ 1.1 Konstruktion der Standardkonfiguration. Construction of the standard configuration.

inertial frames O and O' with Cartesian system of coordinates (for space) in both frames
inertial frame O: space-time coordinates (t, x, y, z), velocity of O' measured in O: v
inertial frame O': space-time coordinates (t', x', y', z'), velocity of O measured in O': v'

Lorentz transformation: (t', x', y', z') = f (t, x, y, z)

We have to find the exact form of the function f !

A) homogeneity of space-time ⇒ space-time transformations between inertial frames are linear. This implies:

$$\begin{pmatrix} t' \\ x' \\ y' \\ z' \end{pmatrix} = f(t, x, y, z) = \begin{pmatrix} \Lambda_{00} & \Lambda_{01} & \Lambda_{02} & \Lambda_{03} \\ \Lambda_{10} & \Lambda_{11} & \Lambda_{12} & \Lambda_{13} \\ \Lambda_{20} & \Lambda_{21} & \Lambda_{22} & \Lambda_{23} \\ \Lambda_{30} & \Lambda_{31} & \Lambda_{32} & \Lambda_{33} \end{pmatrix} . \begin{pmatrix} t \\ x \\ y \\ z \end{pmatrix} + \begin{pmatrix} \alpha_0 \\ \alpha_1 \\ \alpha_2 \\ \alpha_3 \end{pmatrix} \qquad \text{(Poincaré-Transformation)}$$

```
Remove["Global`*"]
```

$$\texttt{L = Table[}\Lambda\texttt{[i, j], \{i, 0, 3\}, \{j, 0, 3\}]; Event = } \begin{pmatrix} t \\ x \\ y \\ z \end{pmatrix}\texttt{; A = Table[}\alpha\texttt{[i], \{i, 0, 3\}];}$$

```
MatrixForm /@ {L.Event + A}
```

$$\left\{ \begin{pmatrix} \alpha[0] + t\,\Lambda[0, 0] + x\,\Lambda[0, 1] + y\,\Lambda[0, 2] + z\,\Lambda[0, 3] \\ \alpha[1] + t\,\Lambda[1, 0] + x\,\Lambda[1, 1] + y\,\Lambda[1, 2] + z\,\Lambda[1, 3] \\ \alpha[2] + t\,\Lambda[2, 0] + x\,\Lambda[2, 1] + y\,\Lambda[2, 2] + z\,\Lambda[2, 3] \\ \alpha[3] + t\,\Lambda[3, 0] + x\,\Lambda[3, 1] + y\,\Lambda[3, 2] + z\,\Lambda[3, 3] \end{pmatrix} \right\}$$

B) Assume the coincidence of space-time origins of the two frames. ⇒ $\alpha_i = 0$

$$\begin{pmatrix} t' \\ x' \\ y' \\ z' \end{pmatrix} = \begin{pmatrix} \Lambda_{00} & \Lambda_{01} & \Lambda_{02} & \Lambda_{03} \\ \Lambda_{10} & \Lambda_{11} & \Lambda_{12} & \Lambda_{13} \\ \Lambda_{20} & \Lambda_{21} & \Lambda_{22} & \Lambda_{23} \\ \Lambda_{30} & \Lambda_{31} & \Lambda_{32} & \Lambda_{33} \end{pmatrix} . \begin{pmatrix} t \\ x \\ y \\ z \end{pmatrix} \qquad \text{(Lorentz-Transformation)}$$

$$\left\{ \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix}, \texttt{"="}, \texttt{L.} \begin{pmatrix} 0 \\ 0 \\ 0 \\ 0 \end{pmatrix} + \texttt{A} \right\}$$

```
{{{0}, {0}, {0}, {0}}, =, {{α[0]}, {α[1]}, {α[2]}, {α[3]}}}
```

C) Assume that at t = 0 the axis of O overlap the axis of O'. ⇒
(0, x, 0, 0) → (t', x', 0, 0) and (0, 0, y, 0) → (t', 0, y', 0) and
(0, 0, 0, z) → (t', 0, 0, z')
⇒ transformation matrix must be like this:

$$\texttt{Lsc} = \begin{pmatrix} \Lambda[0,0] & \Lambda[0,1] & \Lambda[0,2] & \Lambda[0,3] \\ \Lambda[1,0] & \Lambda[1,1] & 0 & 0 \\ \Lambda[2,0] & 0 & \Lambda[2,2] & 0 \\ \Lambda[3,0] & 0 & 0 & \Lambda[3,3] \end{pmatrix};$$

$$\texttt{MatrixForm /@}\left\{\texttt{L}.\begin{pmatrix}0\\x\\0\\0\end{pmatrix}, \texttt{L}.\begin{pmatrix}0\\0\\y\\0\end{pmatrix}, \texttt{L}.\begin{pmatrix}0\\0\\0\\z\end{pmatrix}\right\}$$

$$\texttt{MatrixForm /@}\left\{\texttt{Lsc}.\begin{pmatrix}0\\x\\0\\0\end{pmatrix}, \texttt{Lsc}.\begin{pmatrix}0\\0\\y\\0\end{pmatrix}, \texttt{Lsc}.\begin{pmatrix}0\\0\\0\\z\end{pmatrix}, \texttt{Lsc.Event}\right\}$$

$$\left\{\begin{pmatrix} x\Lambda[0,1] \\ x\Lambda[1,1] \\ x\Lambda[2,1] \\ x\Lambda[3,1] \end{pmatrix}, \begin{pmatrix} y\Lambda[0,2] \\ y\Lambda[1,2] \\ y\Lambda[2,2] \\ y\Lambda[3,2] \end{pmatrix}, \begin{pmatrix} z\Lambda[0,3] \\ z\Lambda[1,3] \\ z\Lambda[2,3] \\ z\Lambda[3,3] \end{pmatrix}\right\}$$

$$\left\{\begin{pmatrix} x\Lambda[0,1] \\ x\Lambda[1,1] \\ 0 \\ 0 \end{pmatrix}, \begin{pmatrix} y\Lambda[0,2] \\ 0 \\ y\Lambda[2,2] \\ 0 \end{pmatrix}, \begin{pmatrix} z\Lambda[0,3] \\ 0 \\ 0 \\ z\Lambda[3,3] \end{pmatrix}, \begin{pmatrix} t\Lambda[0,0] + x\Lambda[0,1] + y\Lambda[0,2] + z\Lambda[0,3] \\ t\Lambda[1,0] + x\Lambda[1,1] \\ t\Lambda[2,0] + y\Lambda[2,2] \\ t\Lambda[3,0] + z\Lambda[3,3] \end{pmatrix}\right\}$$

Note: sc = standard configuration

▪ 1.2 Erstes Relativitätsargument. First relativity argument.

We choose the x-axis of O along v.

D) Two frames obtained by the substitutions y → -z and z → y in O and y' → -z' and z' → y' in O' are connected by the same transformation matrix. This implies:

$$\texttt{yzRotation} = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & -1 \\ 0 & 0 & 1 & 0 \end{pmatrix};$$

```
Lsc = Lsc /. (Reduce[yzRotation.Lsc == Lsc.yzRotation, Flatten[L]] /. {Equal → Rule, And → List});
MatrixForm /@ {%, Lsc.Event}
```

$$\left\{\begin{pmatrix} \Lambda[0,0] & \Lambda[0,1] & 0 & 0 \\ \Lambda[1,0] & \Lambda[1,1] & 0 & 0 \\ 0 & 0 & \Lambda[2,2] & 0 \\ 0 & 0 & 0 & \Lambda[2,2] \end{pmatrix}, \begin{pmatrix} t\Lambda[0,0] + x\Lambda[0,1] \\ t\Lambda[1,0] + x\Lambda[1,1] \\ y\Lambda[2,2] \\ z\Lambda[2,2] \end{pmatrix}\right\}$$

E) block structure of the transformation matrix for standard configuration ⇒ 1D-space case (t, x) can be treated separately from full 3D-space case (t, x, y, z).

The 1D-space case is best done *à la Lévy-Leblond*, see the next subsection.

```
Clear[L]
```

1.3 Jean-Marc Lévy-Leblond, "One more derivation of the Lorentz transformation", American Journal of Physics 44, 271-277 (1976).

■ PRELIMINARIES

"The *principle* of relativity is first stated in general terms, leading to the idea of equivalent frames of reference connected through "inertial" transformations obeying a group law. The *theory* of relativity then is constructed by constraining the transformations through four successive hypotheses: homogeneity of space-time, isotropy of space-time, group structure, causality condition. Only the Lorentz transformations and their degenerate Galilean limit obey these constraints."

"Relativity theory, in fact, is but the statement that all laws of physics are invariant under the Poincaré group (inhomogeneous Lorentz group)."

"I will take as a starting point the statement of the *principle of relativity* in a very general form: there exists an infinite continuous class of reference frames in space-time which are physically equivalent."

"Because of well-known physical considerations, I find it convenient to call "inertial frames" and "inertial transformations" the equivalent reference frames and the transformations connecting them. Indeed, the very existence of such equivalent reference frames corresponds to the validity of the principle of inertia, namely, that a physical object has no absolute state of motion or rest; for instance, a free body (with no "forces" acting on it) is characterized by an "inertial motion" which is not entirely determined, since it depends on "initial conditions", that is also to say, on the reference frame considered."

Let us call "inertial motions" those motions which are obtained from rest by an inertial transformation; an object has an inertial motion in some reference frame if it is at rest in another equivalent frame.

Number of independent parameters in the transformation formulas for the spatiotemporal coordinates (x, t) of an arbitrary event viewed from inertial frames:

Transformation parameters: $a_1 \ldots a_N$

$\{x' = f[x, t, a_1 \ldots a_N], t' = g[x, t, a_1 \ldots a_N]\}$, *N parameters in general*

$\{x' = F[x, t, a_1 \ldots a_n], t' = G[x, t, a_1 \ldots a_n]\}$,
$n = N - 2$ parameters (considering common space - time origins)

$\{x' = F[x, t, a], t' = G[x, t, a]\}$, *dependence on only one parameter (physical considerations)*

■ HYPOTHESIS 1: HOMOGENEITY OF SPACE-TIME

```
<< Utilities`Notation`

Symbolize[x']; Symbolize[t']; Symbolize[xn']; Symbolize[tn'];
```

The direct approach doesn't work yet...

```
☹ DSolve[{
    ∂x F[x, t, a] == H[a], ∂t F[x, t, a] == -K[a],
    ∂x G[x, t, a] == -M[a], ∂t G[x, t, a] == L[a],
    F[0, 0, a] == 0, G[0, 0, a] == 0},
   {F[x, t, a], G[x, t, a]}, {x, t}]
```

$\mathrm{DSolve}[\{F^{(1,0,0)}[x, t, a] == H[a], F^{(0,1,0)}[x, t, a] == -K[a], G^{(1,0,0)}[x, t, a] == -M[a],$
$G^{(0,1,0)}[x, t, a] == L[a], F[0, 0, a] == 0, G[0, 0, a] == 0\}, \{F[x, t, a], G[x, t, a]\}, \{x, t\}]$

...but the indirect approach leads to the wanted result:

```
DSolve[∂x F[x, t, a] == H[a], F[x, t, a], {x, t}];
DSolve[∂t (%[[1, 1, 2]]) == -K[a], C[1][t], t, GeneratedParameters → (Module[{C}, C] &)];
x' = %%[[1, 1, 2]] /. %[[1]];
DSolve[∂x G[x, t, a] == -M[a], G[x, t, a], {x, t}];
DSolve[∂t (%[[1, 1, 2]]) == L[a], C[1][t], t, GeneratedParameters → (Module[{C}, C] &)];
{x', t' = %%[[1, 1, 2]] /. %[[1]]}
Solve[{x' == 0, t' == 0, x == 0, t == 0}, {x'[[1]], t'[[1]]}];
{x', t'} = {x', t'} /. %[[1]]
SolveAlways[{x' == γ (x - ν t), t' == γ (λ t - μ x), v == K[a] / H[a]}, {x, t}]
%[[2]] /. {Rule → Equal, γ → γ[v], λ → λ[v], μ → μ[v], ν → ν[v]} // Flatten;
Solve[%, {H[a], K[a], L[a], M[a]}];
Print["{x',t'} = ", A = {x', t'} = {x', t'} /. %[[1]] // Simplify]
```

```
{C$104 + x H[a] - t K[a], C$128 + t L[a] - x M[a]}

{x H[a] - t K[a], t L[a] - x M[a]}

{{L[a] → 0, M[a] → 0, K[a] → 0, H[a] → 0, γ → 0}, {L[a] → γ λ, M[a] → γ μ, K[a] → v γ, ν → v, H[a] → γ}}

{x',t'} = {(-t v + x) γ[v], γ[v] (t λ[v] - x μ[v])}
```

HYPOTHESIS 2: ISOTROPY OF SPACE

```
B = {xn', tn'} = {-x', t'} /. {x → -x, v → u} // Simplify;
A == B // ExpandAll
({Coefficient[A, x], Coefficient[A, t]} // Flatten) ==
  ({Coefficient[B, x], Coefficient[B, t]} // Flatten);
eqn = Table[%[[1, i]] == %[[2, i]], {i, 1, Length[%[[1]]]}];
% // TableForm
Solve[eqn, u]
eqn = eqn /. % // Flatten
r = eqn[[1]] /. Equal → Rule;
{eqn[[1]], eqn[[4]] /. r, eqn[[2]] /. r};
% // TableForm // FullSimplify
```

```
{-t v γ[v] + x γ[v], t γ[v] λ[v] - x γ[v] μ[v]} == {t u γ[u] + x γ[u], t γ[u] λ[u] + x γ[u] μ[u]}

γ[v] == γ[u]
-γ[v] μ[v] == γ[u] μ[u]
-v γ[v] == u γ[u]
γ[v] λ[v] == γ[u] λ[u]

{{u → -v}}

{γ[v] == γ[-v], -γ[v] μ[v] == γ[-v] μ[-v], -v γ[v] == -v γ[-v], γ[v] λ[v] == γ[-v] λ[-v]}

γ[-v] == γ[v]
γ[-v] (λ[-v] - λ[v]) == 0
γ[-v] (μ[-v] + μ[v]) == 0
```

■ HYPOTHESIS 3: THE GROUP LAW

(a) Identity transformation

```
SolveAlways[{x, t} == {(-t w + x) γ[v], γ[v] (t λ[v] - x μ[v])}, {x, t}] /. v → w
```

```
{{w → 0, μ[w] → 0, λ[w] → 1, γ[w] → 1}}
```

(b) Inverse transformation

```
A = {x == (-t' w + x') γ[ww], t == γ[ww] (t' λ[ww] - x' μ[ww])};
B = Solve[{x', t'} == {(-t v + x) γ[vv], γ[vv] (t λ[vv] - x μ[vv])}, {x, t}] // Flatten;
SolveAlways[{A[[1, 2]] == B[[1, 2]], A[[2, 2]] == B[[2, 2]]}, {x', t'}] /. Rule → Equal;
Flatten[Table[Solve[%[[i]], {w, λ[ww], μ[ww], γ[ww]}], {i, 1, 3}], 1] /. {vv → v, ww → w} // MatrixForm
%[[2]] /. {Rule → Equal, λ[v] → 1}
%[[4]] /. w → -v /. γ[-v] → γ[v]
```

$$\begin{pmatrix} \mu[w]\to 0 & w\to -\frac{v}{\lambda[v]} & \lambda[w]\to \frac{1}{\lambda[v]} & \gamma[w]\to \frac{1}{\gamma[v]} \\ w\to -\frac{v}{\lambda[v]} & \mu[w]\to -\frac{\mu[v]}{\lambda[v]} & \lambda[w]\to \frac{1}{\lambda[v]} & \gamma[w]\to \frac{\lambda[v]}{\gamma[v]\,(\lambda[v]-v\,\mu[v])} \\ w\to 0 & \lambda[w]\to \frac{1}{\lambda[v]} & \mu[w]\to -\frac{\mu[v]}{\lambda[v]} & \gamma[w]\to \frac{1}{\gamma[v]} \end{pmatrix}$$

$$\left\{w == -v,\ \mu[w] == -\mu[v],\ \lambda[w] == 1,\ \gamma[w] == \frac{1}{\gamma[v]\ (1 - v\,\mu[v])}\right\}$$

$$\gamma[v] == \frac{1}{\gamma[v]\ (1 - v\,\mu[v])}$$

λ[v] → 1: See Luigi Masciovecchio "Lévy-Leblond ab omni naevo vindicatus" for a proof that λ(v) = 1.

(c) Composition law

```
Solve[{
   x1 == γ[v1] (x - v1 t), t1 == γ[v1] (t - μ[v1] x),
   x2 == γ[v2] (x1 - v2 t1), t2 == γ[v2] (t1 - μ[v2] x1)},
  {x2, t2}, {x1, t1}] // Flatten
V = -Coefficient[%[[1, 2]], t] / Coefficient[%[[1, 2]], x] // Simplify
Solve[Coefficient[%%[[1, 2]], x] == Coefficient[%%[[2, 2]], t], μ[v1]]
```

```
{x2 → -t v1 γ[v1] γ[v2] - t v2 γ[v1] γ[v2] + x γ[v1] γ[v2] + v2 x γ[v1] γ[v2] μ[v1],
 t2 → t γ[v1] γ[v2] - x γ[v1] γ[v2] μ[v1] + t v1 γ[v1] γ[v2] μ[v2] - x γ[v1] γ[v2] μ[v2]}
```

$$\frac{v1 + v2}{1 + v2\,\mu[v1]}$$

$$\left\{\left\{\mu[v1] \to \frac{v1\,\mu[v2]}{v2}\right\}\right\}$$

```
SolveAlways[μ[v1] / v1 == μ[v2] / v2, {v1, v2}, InverseFunctions → False]
```

$$\text{SolveAlways}\left[\frac{\mu[v1]}{v1} == \frac{\mu[v2]}{v2},\ \{v1, v2\},\ \text{InverseFunctions} \to \text{False}\right]$$

```
{V = V /. μ[v_] → α v,
 eq = γv == 1/(γv (1 - v μ[v])) /. μ[v] → α v}
Reduce[{eq, γv ≥ 0}, γv, Reals]
{"1) Non-causal:", Reduce[{% /. α → -κ^-2, κ > 0}, γv, Reals], V /. α → -κ^-2,
  "2) Galilean:", Reduce[{% /. α → 0}, γv, Reals], V /. α → 0,
  "3) Lorentzian:", Reduce[{% /. α → c^-2, c > 0}, γv, Reals], V /. α → c^-2} // TableForm
```

$$\left\{\frac{v1+v2}{1+v1\,v2\,\alpha},\ \gamma v == \frac{1}{\left(1-v^2\,\alpha\right)\,\gamma v}\right\}$$

$$\left(\alpha \le 0\,\&\&\,\gamma v == \sqrt{\frac{1}{1-v^2\,\alpha}}\right)\ ||\ \left(\alpha > 0\,\&\&\,-\sqrt{\frac{1}{\alpha}} < v < \sqrt{\frac{1}{\alpha}}\,\&\&\,\gamma v == \sqrt{\frac{1}{1-v^2\,\alpha}}\right)$$

1) *Non-causal:*

$$\kappa > 0\,\&\&\,\gamma v == \sqrt{\frac{\kappa^2}{v^2+\kappa^2}}$$

$$\frac{v1+v2}{1-\frac{v1\,v2}{\kappa^2}}$$

2) *Galilean:*

$\gamma v == 1$

$v1 + v2$

3) *Lorentzian:*

$$\left((v \le 0\,\&\&\,c > -v)\ ||\ (v > 0\,\&\&\,c > v)\right)\,\&\&\,\gamma v == \sqrt{\frac{c^2}{c^2-v^2}}$$

$$\frac{v1+v2}{1+\frac{v1\,v2}{c^2}}$$

■ HYPOTHESIS 4: CAUSALITY

```
Xnc[x_, t_] := (1 + v^2/κ^2)^(-1/2) (x - v t)
Tnc[x_, t_] := (1 + v^2/κ^2)^(-1/2) (t + v x/κ^2)
Tnc[x2, t2] - Tnc[x1, t1] // FullSimplify;
{"Non-causal: Δt' = ", % /. (-t1 + t2) → Δt /. (-x1 + x2) → Δx}
```

$$\left\{\text{Non-causal: } \Delta t' = ,\ \frac{v\,\Delta x + \Delta t\,\kappa^2}{\sqrt{1+\frac{v^2}{\kappa^2}}\ \kappa^2}\right\}$$

The sign of the time interval is arbitrary in the non-causal case.

```
Print["Galilean: Δt' = Δt, Δx' = Δx"]
```

Galilean: Δt' = Δt, Δx' = Δx

The sign of the time interval never changes in the Galilean case.

```
Xl[x_, t_] := (1 - v^2 / c^2)^(-1/2) (x - v t)
Tl[x_, t_] := (1 - v^2 / c^2)^(-1/2) (t - v x / c^2)
Tl[x2, t2] - Tl[x1, t1] // FullSimplify;
Δt ' = % /. (-t1 + t2) → Δt /. (x1 - x2) → -Δx;
Xl[x2, t2] - Xl[x1, t1] // Simplify;
Δx ' = % /. v t1 - v t2 → -v Δt /. -(x1 - x2) → Δx;

{"Lorentzian: Δt ' = ", Δt ', " Δx ' = ", Δx '}
```

$$\left\{\text{Lorentzian: } \Delta t' = , \frac{c^2\,\Delta t - v\,\Delta x}{c^2\sqrt{1-\frac{v^2}{c^2}}},\ \Delta x' = , \frac{-v\,\Delta t + \Delta x}{\sqrt{1-\frac{v^2}{c^2}}}\right\}$$

For | Δx/Δt | < c (time-like intervals) the sign of the time interval doesn't change in the Lorentzian case.

■ 1.4 Zweites Relativitätsargument. Second relativity argument.

From the preceding section we have:

```
(Λ[0, 0]  Λ[0, 1]
 Λ[1, 0]  Λ[1, 1]) = γ (1   -v / c^2
                        -v   1      );
Lsc // MatrixForm
```

$$\begin{pmatrix} \gamma & -\frac{v\gamma}{c^2} & 0 & 0 \\ -v\gamma & \gamma & 0 & 0 \\ 0 & 0 & \Lambda[2,2] & 0 \\ 0 & 0 & 0 & \Lambda[2,2] \end{pmatrix}$$

F) Two frames obtained by the substitutions x→-x and z→-z in O and x'→-x' and z'→-z' in O' are connected by the same transformation matrix **Λ**, we have proved that v = - v' (reciprocity), we have also the continuity of **Λ**(v) and **Λ**(v=0) = **1** (identity). This implies:

```
xzInvertion = DiagonalMatrix[{1, -1, 1, -1}];
xzInvertion == Lsc.xzInvertion.Lsc /. γ → (1 - v^2 / c^2)^(-1/2) // Simplify
Solve[%]
Lsc = Lsc /. %[[2]];
```

$\Lambda[2, 2]^2 == 1$

{{Λ[2, 2] → -1}, {Λ[2, 2] → 1}}

The Lorentz transformation for frames in standard configuration are finally:

```
MatrixForm /@ {Lsc, Lsc.Event} // Simplify
```

$$\left\{\begin{pmatrix} \gamma & -\frac{v\gamma}{c^2} & 0 & 0 \\ -v\gamma & \gamma & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} \left(t - \frac{v x}{c^2}\right)\gamma \\ (-t v + x)\gamma \\ y \\ z \end{pmatrix}\right\}$$

■ 1.A Rechenbeispiele. Examples of calculations.

```
SGT[x_, y_, z_, t_, v_] := {x - v t, y, z, t}; VGAdd[v1_, v2_] := v1 + v2;
SLT[x_, y_, z_, t_, v_] := (ɣ := (1 - v^2 / c^2)^(-1/2); {(-t v + x) ɣ, y, z, (t - v x / c^2) ɣ})
VLAdd[v1_, v2_] := (v1 + v2) / (1 + v1 v2 / c^2)
```

```
c = 1; {SLT[4, 2, 3, 9, .6 c], VLAdd[.5, .6]}
```

```
{{-1.75, 2, 3, 8.25}, 0.846154}
```

```
(*T.Hebbeker, 7. P2 TH 02 Relativitätstheorie I*)
c = 299 792 458 / 1000 (*km/s*);
SGT[100, 100, 100, 0.01 , .1 c]
{SLT[100, 100, 100, 0.01 , .1 c], VLAdd[.99 c, .90 c] / c}
MapAt[N, SLT[(2 / 1000), 0, 0, (866 / 1000) / c (2 / 1000), (866 / 1000) c], 1]
SLT[0, 0, 0, 10, .943 c]
```

```
{-199.792, 100, 100, 0.01}
```

```
{{-200.799, 100, 100, 0.0100169}, 0.999471}
```

```
{0.00100009, 0, 0, 0}
```

$\{-8.49491\times 10^{6}, 0, 0, 30.0487\}$

```
Clear[c]
{SLT[Δl, 0, 0, v / c^2 Δl, v], SLT[0, 0, 0, 0, v]} // Simplify
{SLT[0, 0, 0, Δt, v], SLT[0, 0, 0, 0, v]} // Simplify
```

$$\left\{\left\{\sqrt{1-\frac{v^2}{c^2}}\,\Delta l, 0, 0, 0\right\}, \{0, 0, 0, 0\}\right\}$$

$$\left\{\left\{-\frac{v\,\Delta t}{\sqrt{1-\frac{v^2}{c^2}}}, 0, 0, \frac{\Delta t}{\sqrt{1-\frac{v^2}{c^2}}}\right\}, \{0, 0, 0, 0\}\right\}$$

2. Allgemeine Lorentztransformation (Lorentz transformation for frames in arbitrary configuration)

■ 2.1 Darstellungen. Representations.

```
Clear[ɣ]
```

```
gr = ɣ → (1 - β^2)^(-1/2);
grC = {ɣ → (1 - (vx^2 + vy^2 + vz^2) / c^2)^(-1/2), v → Sqrt[vx^2 + vy^2 + vz^2]};
```

I. Spezielle Lorentztransformation. Koordinatensysteme in Standardkonfiguration. Relative Geschwindigkeit $\vec{v} = (v, 0, 0)$.
Lorentz transformation for frames in standard configuration.

$$\Lambda slt = \begin{pmatrix} \gamma & -\frac{\beta\gamma}{c} & 0 & 0 \\ -\beta c \gamma & \gamma & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}; \; Event = \begin{pmatrix} t \\ x \\ y \\ z \end{pmatrix};$$

```
Λslt /. gr /. β → v / c // MatrixForm
```

$$\begin{pmatrix} \frac{1}{\sqrt{1-\frac{v^2}{c^2}}} & -\frac{v}{c^2\sqrt{1-\frac{v^2}{c^2}}} & 0 & 0 \\ -\frac{v}{\sqrt{1-\frac{v^2}{c^2}}} & \frac{1}{\sqrt{1-\frac{v^2}{c^2}}} & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

II. Allgemeine Lorentztransformation für Koordinatensysteme "ohne" relative Drehung.
Relative Geschwindigkeit $\vec{v}$ durch $\{\phi v, \theta v, v\}$ oder durch $\vec{v} = (vx, vy, vz)$ gegeben.
Lorentz transformation for frames "without" relative rotation of the axis and moving with relative velocity $\vec{v}$.

$$Rv = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & \mathrm{Cos}[\phi v] & 0 & \mathrm{Sin}[\phi v] \\ 0 & 0 & 1 & 0 \\ 0 & -\mathrm{Sin}[\phi v] & 0 & \mathrm{Cos}[\phi v] \end{pmatrix} . \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & \mathrm{Cos}[\theta v] & \mathrm{Sin}[\theta v] & 0 \\ 0 & -\mathrm{Sin}[\theta v] & \mathrm{Cos}[\theta v] & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix};$$

```
InvRv = Inverse[Rv] // Simplify;
MatrixForm /@ {Rv, InvRv}
```

$$\left\{ \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & \mathrm{Cos}[\theta v]\,\mathrm{Cos}[\phi v] & \mathrm{Cos}[\phi v]\,\mathrm{Sin}[\theta v] & \mathrm{Sin}[\phi v] \\ 0 & -\mathrm{Sin}[\theta v] & \mathrm{Cos}[\theta v] & 0 \\ 0 & -\mathrm{Cos}[\theta v]\,\mathrm{Sin}[\phi v] & -\mathrm{Sin}[\theta v]\,\mathrm{Sin}[\phi v] & \mathrm{Cos}[\phi v] \end{pmatrix}, \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & \mathrm{Cos}[\theta v]\,\mathrm{Cos}[\phi v] & -\mathrm{Sin}[\theta v] & -\mathrm{Cos}[\theta v]\,\mathrm{Sin}[\phi v] \\ 0 & \mathrm{Cos}[\phi v]\,\mathrm{Sin}[\theta v] & \mathrm{Cos}[\theta v] & -\mathrm{Sin}[\theta v]\,\mathrm{Sin}[\phi v] \\ 0 & \mathrm{Sin}[\phi v] & 0 & \mathrm{Cos}[\phi v] \end{pmatrix} \right\}$$

```
Λaltod = InvRv.Λslt.Rv // Simplify;
Λaltod[[2, 2]] = Λaltod[[2, 2]] /. {Sin[θv]^2 → (1 - Cos[θv]^2), Sin[ϕv]^2 → (1 - Cos[ϕv]^2)} // Simplify;
Λaltod[[3, 3]] = Λaltod[[3, 3]] /. {Cos[θv]^2 → (1 - Sin[θv]^2), Sin[ϕv]^2 → (1 - Cos[ϕv]^2)} // Simplify;
Λaltod[[4, 4]] = Λaltod[[4, 4]] /. {Cos[ϕv]^2 → (1 - Sin[ϕv]^2)} // Simplify;

AtCr = {Sin[θv] → vy / Sqrt[vx^2 + vy^2], Cos[θv] → vx / Sqrt[vx^2 + vy^2],
   Sin[ϕv] → vz / v, Cos[ϕv] → Sqrt[vx^2 + vy^2] / v, β → v / c};
ΛaltodC = Collect[Λaltod /. AtCr, vz^2 / v^2];

Λaltod // MatrixForm
ΛaltodC // MatrixForm
```

$$\begin{pmatrix} \gamma & -\frac{\beta\gamma\,\mathrm{Cos}[\theta v]\,\mathrm{Cos}[\phi v]}{c} & -\frac{\beta\gamma\,\mathrm{Cos}[\phi v]\,\mathrm{Sin}[\theta v]}{c} & -\frac{\beta\gamma\,\mathrm{Sin}[\phi v]}{c} \\ -c\beta\gamma\,\mathrm{Cos}[\theta v]\,\mathrm{Cos}[\phi v] & 1+(-1+\gamma)\,\mathrm{Cos}[\theta v]^2\,\mathrm{Cos}[\phi v]^2 & (-1+\gamma)\,\mathrm{Cos}[\theta v]\,\mathrm{Cos}[\phi v]^2\,\mathrm{Sin}[\theta v] & (-1+\gamma)\,\mathrm{Cos}[\theta v]\,\mathrm{Cos}[\phi v]\,\mathrm{Sin}[\phi v] \\ -c\beta\gamma\,\mathrm{Cos}[\phi v]\,\mathrm{Sin}[\theta v] & (-1+\gamma)\,\mathrm{Cos}[\theta v]\,\mathrm{Cos}[\phi v]^2\,\mathrm{Sin}[\theta v] & 1+(-1+\gamma)\,\mathrm{Cos}[\phi v]^2\,\mathrm{Sin}[\theta v]^2 & (-1+\gamma)\,\mathrm{Cos}[\phi v]\,\mathrm{Sin}[\theta v]\,\mathrm{Sin}[\phi v] \\ -c\beta\gamma\,\mathrm{Sin}[\phi v] & (-1+\gamma)\,\mathrm{Cos}[\theta v]\,\mathrm{Cos}[\phi v]\,\mathrm{Sin}[\phi v] & (-1+\gamma)\,\mathrm{Cos}[\phi v]\,\mathrm{Sin}[\theta v]\,\mathrm{Sin}[\phi v] & 1+(-1+\gamma)\,\mathrm{Sin}[\phi v]^2 \end{pmatrix}$$

$$\begin{pmatrix} \gamma & -\frac{vx\,\gamma}{c^2} & -\frac{vy\,\gamma}{c^2} & -\frac{vz\,\gamma}{c^2} \\ -vx\,\gamma & 1+\frac{vx^2\,(-1+\gamma)}{v^2} & \frac{vx\,vy\,(-1+\gamma)}{v^2} & \frac{vx\,vz\,(-1+\gamma)}{v^2} \\ -vy\,\gamma & \frac{vx\,vy\,(-1+\gamma)}{v^2} & 1+\frac{vy^2\,(-1+\gamma)}{v^2} & \frac{vy\,vz\,(-1+\gamma)}{v^2} \\ -vz\,\gamma & \frac{vx\,vz\,(-1+\gamma)}{v^2} & \frac{vy\,vz\,(-1+\gamma)}{v^2} & 1+\frac{vz^2\,(-1+\gamma)}{v^2} \end{pmatrix}$$

III. Allgemeine Lorentztransformation für Koordinatensysteme "mit" relativer Drehung.

Relative Drehung durch $\{\psi, \theta, \phi\}$ gegeben. Relative Geschwindigkeit $\vec{v}$ durch $\{\phi v, \theta v, v\}$ oder durch $\vec{v}$ = (vx, vy, vz) gegeben.

Lorentz transformation for frames "with" relative rotation of the axis given by the rotation matrix R and moving with relative velocity $\vec{v}$.

$$R = \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & \text{Cos}[\psi] & \text{Sin}[\psi] & 0 \\ 0 & -\text{Sin}[\psi] & \text{Cos}[\psi] & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix} . \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & \text{Cos}[\theta] & \text{Sin}[\theta] \\ 0 & 0 & -\text{Sin}[\theta] & \text{Cos}[\theta] \end{pmatrix} . \begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & \text{Cos}[\phi] & \text{Sin}[\phi] & 0 \\ 0 & -\text{Sin}[\phi] & \text{Cos}[\phi] & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix};$$

```
Λaltmd = R.Λaltod // Simplify;
For[i = 1, i ≤ 4, i++,
 Print["\n", i, ". column of the tranformation matrix:\n", MatrixForm[Transpose[Λaltmd][[i]]]]]
```

1. column of the tranformation matrix:

$$\begin{pmatrix} \gamma \\ -c\,\beta\,\gamma\,(\text{Sin}[\theta]\,\text{Sin}[\phi v]\,\text{Sin}[\psi] + \text{Cos}[\phi v]\,\text{Sin}[\theta v]\,(\text{Cos}[\psi]\,\text{Sin}[\phi] + \text{Cos}[\theta]\,\text{Cos}[\phi]\,\text{Sin}[\psi]) + \text{Cos}[\theta v]\,\text{Cos}[\phi v]\,(\text{Cos}[\phi]\,\text{Cos}[\psi] - \text{Cos}[\theta]\,\text{Sin}[\phi]\,\text{Sin}[\psi])) \\ -c\,\beta\,\gamma\,(\text{Cos}[\theta]\,\text{Cos}[\phi v]\,\text{Cos}[\psi]\,\text{Sin}[\theta v - \phi] + \text{Cos}[\psi]\,\text{Sin}[\theta]\,\text{Sin}[\phi v] - \text{Cos}[\theta v - \phi]\,\text{Cos}[\phi v]\,\text{Sin}[\psi]) \\ c\,\beta\,\gamma\,(\text{Cos}[\phi]\,\text{Cos}[\phi v]\,\text{Sin}[\theta]\,\text{Sin}[\theta v] - \text{Cos}[\theta v]\,\text{Cos}[\phi v]\,\text{Sin}[\theta]\,\text{Sin}[\phi] - \text{Cos}[\theta]\,\text{Sin}[\phi v]) \end{pmatrix}$$

2. column of the tranformation matrix:

$$\begin{pmatrix} -\frac{\beta\,\gamma\,\text{Cos}[\theta v]\,\text{Cos}[\phi v]}{c} \\ (-1+\gamma)\,\text{Cos}[\theta v]\,\text{Cos}[\phi v]\,\text{Sin}[\theta]\,\text{Sin}[\phi v]\,\text{Sin}[\psi] + (-1+\gamma)\,\text{Cos}[\theta v]\,\text{Cos}[\phi v]^2\,\text{Sin}[\theta v]\,(\text{Cos}[\psi]\,\text{Sin}[\phi] + \text{Cos}[\theta]\,\text{Cos}[\phi]\,\text{Sin}[\psi]) + (1 + (-1+\gamma)\,\text{Cos}[\theta v]^2\,\text{Cos}[\phi v]^2)\,(\text{Cos}[\phi]\,\text{Cos}[\psi] - \text{Cos}[\theta]\,\text{Sin}[\phi]\,\text{Sin}[\psi]) \\ (-1+\gamma)\,\text{Cos}[\theta v]\,\text{Cos}[\phi v]\,\text{Cos}[\psi]\,\text{Sin}[\theta]\,\text{Sin}[\phi v] + (1 + (-1+\gamma)\,\text{Cos}[\theta v]^2\,\text{Cos}[\phi v]^2)\,(-\text{Cos}[\theta]\,\text{Cos}[\psi]\,\text{Sin}[\phi] - \text{Cos}[\phi]\,\text{Sin}[\psi]) + (-1+\gamma)\,\text{Cos}[\theta v]\,\text{Cos}[\phi v]^2\,\text{Sin}[\theta v]\,(\text{Cos}[\theta]\,\text{Cos}[\phi]\,\text{Cos}[\psi] - \text{Sin}[\phi]\,\text{Sin}[\psi]) \\ \text{Sin}[\theta]\,\text{Sin}[\phi] + (-1+\gamma)\,\text{Cos}[\theta v]^2\,\text{Cos}[\phi v]^2\,\text{Sin}[\theta]\,\text{Sin}[\phi] - (-1+\gamma)\,\text{Cos}[\theta v]\,\text{Cos}[\phi v]\,(\text{Cos}[\phi]\,\text{Cos}[\phi v]\,\text{Sin}[\theta]\,\text{Sin}[\theta v] - \text{Cos}[\theta]\,\text{Sin}[\phi v]) \end{pmatrix}$$

3. column of the tranformation matrix:

$$\begin{pmatrix} -\frac{\beta\,\gamma\,\text{Cos}[\phi v]\,\text{Sin}[\theta v]}{c} \\ (-1+\gamma)\,\text{Cos}[\phi v]\,\text{Sin}[\theta]\,\text{Sin}[\theta v]\,\text{Sin}[\phi v]\,\text{Sin}[\psi] + (1 + (-1+\gamma)\,\text{Cos}[\phi v]^2\,\text{Sin}[\theta v]^2)\,(\text{Cos}[\psi]\,\text{Sin}[\phi] + \text{Cos}[\theta]\,\text{Cos}[\phi]\,\text{Sin}[\psi]) + (-1+\gamma)\,\text{Cos}[\theta v]\,\text{Cos}[\phi v]^2\,\text{Sin}[\theta v]\,(\text{Cos}[\phi]\,\text{Cos}[\psi] - \text{Cos}[\theta]\,\text{Sin}[\phi]\,\text{Sin}[\psi]) \\ (-1+\gamma)\,\text{Cos}[\phi v]\,\text{Cos}[\psi]\,\text{Sin}[\theta]\,\text{Sin}[\theta v]\,\text{Sin}[\phi v] - (-1+\gamma)\,\text{Cos}[\theta v]\,\text{Cos}[\phi v]^2\,\text{Sin}[\theta v]\,(\text{Cos}[\theta]\,\text{Cos}[\psi]\,\text{Sin}[\phi] + \text{Cos}[\phi]\,\text{Sin}[\psi]) + (1 + (-1+\gamma)\,\text{Cos}[\phi v]^2\,\text{Sin}[\theta v]^2)\,(\text{Cos}[\theta]\,\text{Cos}[\phi]\,\text{Cos}[\psi] - \text{Sin}[\phi]\,\text{Sin}[\psi]) \\ -\text{Cos}[\phi]\,\text{Sin}[\theta]\,(1 + (-1+\gamma)\,\text{Cos}[\phi v]^2\,\text{Sin}[\theta v]^2) + (-1+\gamma)\,\text{Cos}[\phi v]\,\text{Sin}[\theta v]\,(\text{Cos}[\theta v]\,\text{Cos}[\phi v]\,\text{Sin}[\theta]\,\text{Sin}[\phi] + \text{Cos}[\theta]\,\text{Sin}[\phi v]) \end{pmatrix}$$

4. column of the tranformation matrix:

$$\begin{pmatrix} -\frac{\beta\,\gamma\,\text{Sin}[\phi v]}{c} \\ \text{Sin}[\theta]\,(1 + (-1+\gamma)\,\text{Sin}[\phi v]^2)\,\text{Sin}[\psi] + (-1+\gamma)\,\text{Cos}[\phi v]\,\text{Sin}[\theta v]\,\text{Sin}[\phi v]\,(\text{Cos}[\psi]\,\text{Sin}[\phi] + \text{Cos}[\theta]\,\text{Cos}[\phi]\,\text{Sin}[\psi]) + (-1+\gamma)\,\text{Cos}[\theta v]\,\text{Cos}[\phi v]\,\text{Sin}[\phi v]\,(\text{Cos}[\phi]\,\text{Cos}[\psi] - \text{Cos}[\theta]\,\text{Sin}[\phi]\,\text{Sin}[\psi]) \\ \text{Cos}[\psi]\,\left(\text{Sin}[\theta]\,(1 + (-1+\gamma)\,\text{Sin}[\phi v]^2) + \frac{1}{2}\,(-1+\gamma)\,\text{Cos}[\theta]\,\text{Sin}[\theta v - \phi]\,\text{Sin}[2\,\phi v]\right) - \frac{1}{2}\,(-1+\gamma)\,\text{Cos}[\theta v - \phi]\,\text{Sin}[2\,\phi v]\,\text{Sin}[\psi] \\ \text{Cos}[\theta]\,(1 + (-1+\gamma)\,\text{Sin}[\phi v]^2) - \frac{1}{2}\,(-1+\gamma)\,\text{Sin}[\theta]\,\text{Sin}[\theta v - \phi]\,\text{Sin}[2\,\phi v] \end{pmatrix}$$

```
ΛaltmdC = R.ΛaltodC // Simplify;
For[i = 1, i ≤ 4, i++,
 Print["\n", i, ". column of the tranformation matrix:\n", MatrixForm[Transpose[ΛaltmdC][[i]]]]]
```

1. column of the tranformation matrix:

$$\begin{pmatrix} \gamma \\ -\gamma\,(\text{vy}\,\text{Cos}[\psi]\,\text{Sin}[\phi] + (\text{vz}\,\text{Sin}[\theta] - \text{vx}\,\text{Cos}[\theta]\,\text{Sin}[\phi])\,\text{Sin}[\psi] + \text{Cos}[\phi]\,(\text{vx}\,\text{Cos}[\psi] + \text{vy}\,\text{Cos}[\theta]\,\text{Sin}[\psi])) \\ \gamma\,(-\text{vz}\,\text{Cos}[\psi]\,\text{Sin}[\theta] - \text{Cos}[\theta]\,\text{Cos}[\psi]\,(\text{vy}\,\text{Cos}[\phi] - \text{vx}\,\text{Sin}[\phi]) + (\text{vx}\,\text{Cos}[\phi] + \text{vy}\,\text{Sin}[\phi])\,\text{Sin}[\psi]) \\ -\gamma\,(\text{vz}\,\text{Cos}[\theta] + \text{Sin}[\theta]\,(-\text{vy}\,\text{Cos}[\phi] + \text{vx}\,\text{Sin}[\phi])) \end{pmatrix}$$

2. column of the tranformation matrix:

$$\begin{pmatrix} -\frac{\text{vx}\,\gamma}{c^2} \\ \frac{\text{vx}\,\text{vz}\,(-1+\gamma)\,\text{Sin}[\theta]\,\text{Sin}[\psi]}{v^2} + \frac{\text{vx}\,\text{vy}\,(-1+\gamma)\,(\text{Cos}[\psi]\,\text{Sin}[\phi] + \text{Cos}[\theta]\,\text{Cos}[\phi]\,\text{Sin}[\psi])}{v^2} + \left(1 + \frac{\text{vx}^2\,(-1+\gamma)}{v^2}\right)\,(\text{Cos}[\phi]\,\text{Cos}[\psi] - \text{Cos}[\theta]\,\text{Sin}[\phi]\,\text{Sin}[\psi]) \\ \frac{\text{vx}\,\text{vz}\,(-1+\gamma)\,\text{Cos}[\psi]\,\text{Sin}[\theta]}{v^2} + \left(1 + \frac{\text{vx}^2\,(-1+\gamma)}{v^2}\right)\,(-\text{Cos}[\theta]\,\text{Cos}[\psi]\,\text{Sin}[\phi] - \text{Cos}[\phi]\,\text{Sin}[\psi]) + \frac{\text{vx}\,\text{vy}\,(-1+\gamma)\,(\text{Cos}[\theta]\,\text{Cos}[\phi]\,\text{Cos}[\psi] - \text{Sin}[\phi]\,\text{Sin}[\psi])}{v^2} \\ \frac{\text{vx}\,\text{vz}\,(-1+\gamma)\,\text{Cos}[\theta] + \text{Sin}[\theta]\,(-\text{vx}\,\text{vy}\,(-1+\gamma)\,\text{Cos}[\phi] + (v^2 + \text{vx}^2\,(-1+\gamma))\,\text{Sin}[\phi])}{v^2} \end{pmatrix}$$

3. column of the tranformation matrix:

$$\begin{pmatrix} -\frac{vy\,\gamma}{c^2} \\ \frac{vy\,vz\,(-1+\gamma)\,\mathrm{Sin}[\theta]\,\mathrm{Sin}[\psi]}{v^2} + \left(1+\frac{vy^2\,(-1+\gamma)}{v^2}\right)(\mathrm{Cos}[\psi]\,\mathrm{Sin}[\phi]+\mathrm{Cos}[\theta]\,\mathrm{Cos}[\phi]\,\mathrm{Sin}[\psi]) + \frac{vx\,vy\,(-1+\gamma)\,(\mathrm{Cos}[\phi]\,\mathrm{Cos}[\psi]-\mathrm{Cos}[\theta]\,\mathrm{Sin}[\phi]\,\mathrm{Sin}[\psi])}{v^2} \\ \frac{vy\,vz\,(-1+\gamma)\,\mathrm{Cos}[\psi]\,\mathrm{Sin}[\theta]}{v^2} - \frac{vx\,vy\,(-1+\gamma)\,(\mathrm{Cos}[\theta]\,\mathrm{Cos}[\psi]\,\mathrm{Sin}[\phi]+\mathrm{Cos}[\phi]\,\mathrm{Sin}[\psi])}{v^2} + \left(1+\frac{vy^2\,(-1+\gamma)}{v^2}\right)(\mathrm{Cos}[\theta]\,\mathrm{Cos}[\phi]\,\mathrm{Cos}[\psi]-\mathrm{Sin}[\phi]\,\mathrm{Sin}[\psi]) \\ \frac{vy\,vz\,(-1+\gamma)\,\mathrm{Cos}[\theta]-\mathrm{Sin}[\theta]\,\left(\left(v^2+vy^2\,(-1+\gamma)\right)\mathrm{Cos}[\phi]-vx\,vy\,(-1+\gamma)\,\mathrm{Sin}[\phi]\right)}{v^2} \end{pmatrix}$$

4. column of the tranformation matrix:

$$\begin{pmatrix} -\frac{vz\,\gamma}{c^2} \\ \frac{vy\,vz\,(-1+\gamma)\,\mathrm{Cos}[\psi]\,\mathrm{Sin}[\phi]+\left(\left(v^2+vz^2\,(-1+\gamma)\right)\mathrm{Sin}[\theta]-vx\,vz\,(-1+\gamma)\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\phi]\right)\mathrm{Sin}[\psi]+vz\,(-1+\gamma)\,\mathrm{Cos}[\phi]\,(vx\,\mathrm{Cos}[\psi]+vy\,\mathrm{Cos}[\theta]\,\mathrm{Sin}[\psi])}{v^2} \\ \frac{\left(v^2+vz^2\,(-1+\gamma)\right)\mathrm{Cos}[\psi]\,\mathrm{Sin}[\theta]+vz\,(-1+\gamma)\,\mathrm{Cos}[\theta]\,\mathrm{Cos}[\psi]\,(vy\,\mathrm{Cos}[\phi]-vx\,\mathrm{Sin}[\phi])-vz\,(-1+\gamma)\,(vx\,\mathrm{Cos}[\phi]+vy\,\mathrm{Sin}[\phi])\,\mathrm{Sin}[\psi]}{v^2} \\ \frac{\left(v^2+vz^2\,(-1+\gamma)\right)\mathrm{Cos}[\theta]-vz\,(-1+\gamma)\,\mathrm{Sin}[\theta]\,(vy\,\mathrm{Cos}[\phi]-vx\,\mathrm{Sin}[\phi])}{v^2} \end{pmatrix}$$

2.2 Beweis der Isometrie der Lorentztransformation im Minkowsi-Raum; Fälle I, II und III. Proof of the isometry of the transformation formulas in the Minkowski space; cases I, II and III.

```
Δssquare = {c² t² - x² - y² - z²};
STI := (c EP[[1]])^2 - EP[[2]]^2 - EP[[3]]^2 - EP[[4]]^2;
Print["Case I: Standard configuration."];
EP = Δslt.Event; (STI /. gr // Simplify) == Δssquare
Print["Case II: General configuration 'without' rotation."];
EP = Δaltod.Event; (STI /. gr // Simplify) == Δssquare
EP = ΔaltodC.Event; (STI /. grC // Simplify) == Δssquare
Print["Case III: General configuration 'with' rotation."];
EP = Δaltmd.Event; Timing[( STI /. gr // Simplify) == Δssquare]
EP = ΔaltmdC.Event; Timing[(STI /. grC // Simplify) == Δssquare]
```

```
Case I: Standard configuration.

True

Case II: General configuration 'without' rotation.

True

True

Case III: General configuration 'with' rotation.

{175.33 Second, True}

{102.38 Second, True}
```

In each of the three cases, the previously found Lorentz transformations are **isometric** as shown here by direct verification of the equality $\Delta s'^2 = \Delta s^2 = c^2\,t^2 - x^2 - y^2 - z^2$. Apart from the discrete Lorentz transformations, not considered here, they are the **only** isometric transformations in the Minkowski space (for a proof of this assertion see Steven Weinberg, *Gravitation and Cosmology*, 1972, p.27).

■ 2.A Vergleich der Ergebnisse. Comparison of results.

My formula:

```
a = Λaltod /. gr // MatrixForm;
{a /. {β -> β, ϕv → 0, θv -> 0},
 a /. {β -> β, ϕv → Pi / 2, θv -> 0},
 a /. {β → 0.24, ϕv → 3, θv → 1964}}
```

$$\left\{\begin{pmatrix} \frac{1}{\sqrt{1-\beta^2}} & -\frac{\beta}{c\sqrt{1-\beta^2}} & 0 & 0 \\ -\frac{c\,\beta}{\sqrt{1-\beta^2}} & \frac{1}{\sqrt{1-\beta^2}} & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} \frac{1}{\sqrt{1-\beta^2}} & 0 & 0 & -\frac{\beta}{c\sqrt{1-\beta^2}} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ -\frac{c\,\beta}{\sqrt{1-\beta^2}} & 0 & 0 & \frac{1}{\sqrt{1-\beta^2}} \end{pmatrix},\right.$$

$$\left.\begin{pmatrix} 1.03011 & -\frac{0.214249}{c} & -\frac{0.118325}{c} & -\frac{0.0348885}{c} \\ -0.214249\,c & 1.02261 & 0.0124875 & 0.00368198 \\ -0.118325\,c & 0.0124875 & 1.0069 & 0.00203348 \\ -0.0348885\,c & 0.00368198 & 0.00203348 & 1.0006 \end{pmatrix}\right\}$$

David Park *TGeneralRelativity1`GeneralRelativity`* *Mathematica Tensorial* subpackage (Note: c = 1):

```
Needs["TGeneralRelativity1`GeneralRelativity`"]

$PrePrint =.
DeclareBaseIndices[{0, 1, 2, 3}]
DeclareIndexFlavor[{red, Red}]
DefineTensorShortcuts[{ΛP, 2}]

SetLorentzBoost[ΛP, red, Identity][β, ϕ, θ]
ΛPud[red@α, β] == (ΛPud[red@α, β] // ToArrayValues[]);
Last[%] // MatrixForm
SetLorentzBoost[ΛP, red, Identity][β, 0, 0]
ΛPud[red@α, β] == (ΛPud[red@α, β] // ToArrayValues[]);
a = Last[%] // MatrixForm;
SetLorentzBoost[ΛP, red, Identity][β, π / 2, 0]
ΛPud[red@α, β] == (ΛPud[red@α, β] // ToArrayValues[]);
b = Last[%] // MatrixForm;
SetLorentzBoost[ΛP, red, Identity][0.24, 3, 1964]
ΛPud[red@α, β] == (ΛPud[red@α, β] // ToArrayValues[]);
c = Last[%] // MatrixForm; {a, b, c}
```

$$\begin{pmatrix} \frac{1}{\sqrt{1-\beta^2}} & -\frac{\beta \operatorname{Cos}[\theta] \operatorname{Cos}[\phi]}{\sqrt{1-\beta^2}} & -\frac{\beta \operatorname{Cos}[\phi] \operatorname{Sin}[\theta]}{\sqrt{1-\beta^2}} & -\frac{\beta \operatorname{Sin}[\phi]}{\sqrt{1-\beta^2}} \\ -\frac{\beta \operatorname{Cos}[\theta] \operatorname{Cos}[\phi]}{\sqrt{1-\beta^2}} & 1+\left(-1+\frac{1}{\sqrt{1-\beta^2}}\right) \operatorname{Cos}[\theta]^2 \operatorname{Cos}[\phi]^2 & \left(-1+\frac{1}{\sqrt{1-\beta^2}}\right) \operatorname{Cos}[\theta] \operatorname{Cos}[\phi]^2 \operatorname{Sin}[\theta] & \left(-1+\frac{1}{\sqrt{1-\beta^2}}\right) \operatorname{Cos}[\theta] \operatorname{Cos}[\phi] \operatorname{Sin}[\phi] \\ -\frac{\beta \operatorname{Cos}[\phi] \operatorname{Sin}[\theta]}{\sqrt{1-\beta^2}} & \left(-1+\frac{1}{\sqrt{1-\beta^2}}\right) \operatorname{Cos}[\theta] \operatorname{Cos}[\phi]^2 \operatorname{Sin}[\theta] & 1+\left(-1+\frac{1}{\sqrt{1-\beta^2}}\right) \operatorname{Cos}[\phi]^2 \operatorname{Sin}[\theta]^2 & \left(-1+\frac{1}{\sqrt{1-\beta^2}}\right) \operatorname{Cos}[\phi] \operatorname{Sin}[\theta] \operatorname{Sin}[\phi] \\ -\frac{\beta \operatorname{Sin}[\phi]}{\sqrt{1-\beta^2}} & \left(-1+\frac{1}{\sqrt{1-\beta^2}}\right) \operatorname{Cos}[\theta] \operatorname{Cos}[\phi] \operatorname{Sin}[\phi] & \left(-1+\frac{1}{\sqrt{1-\beta^2}}\right) \operatorname{Cos}[\phi] \operatorname{Sin}[\theta] \operatorname{Sin}[\phi] & 1+\left(-1+\frac{1}{\sqrt{1-\beta^2}}\right) \operatorname{Sin}[\phi]^2 \end{pmatrix}$$

$$\left\{\begin{pmatrix} \frac{1}{\sqrt{1-\beta^2}} & -\frac{\beta}{\sqrt{1-\beta^2}} & 0 & 0 \\ -\frac{\beta}{\sqrt{1-\beta^2}} & \frac{1}{\sqrt{1-\beta^2}} & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} \frac{1}{\sqrt{1-\beta^2}} & 0 & 0 & -\frac{\beta}{\sqrt{1-\beta^2}} \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ -\frac{\beta}{\sqrt{1-\beta^2}} & 0 & 0 & \frac{1}{\sqrt{1-\beta^2}} \end{pmatrix}, \begin{pmatrix} 1.03011 & -0.214249 & -0.118325 & -0.0348885 \\ -0.214249 & 1.02261 & 0.0124875 & 0.00368198 \\ -0.118325 & 0.0124875 & 1.0069 & 0.00203348 \\ -0.0348885 & 0.00368198 & 0.00203348 & 1.0006 \end{pmatrix}\right\}$$

Anhang (Appendix)

Definitions of the SI base units.

The *second* is the duration of 9 192 631 770 periods of the radiation corresponding to the transition between the two hyperfine levels of the ground state of the cesium 133 atom.

The *meter* is the length of the path travelled by light in vacuum during a time interval of 1/299 792 458 of a second.

```
<< Miscellaneous`PhysicalConstants`
{SpeedOfLight}
```

$$\left\{\frac{299\,792\,458\ \text{Meter}}{\text{Second}}\right\}$$

"That's all Folks!"

* * *